Centrum voor Wiskunde en Informatica
Centre for Mathematics and Computer Science

H.J.J. te Riele

A program for solving first kind Fredholm integral equations by means of regularization

Department of Numerical Mathematics Report NM-R8416 November

The Centre for Mathematics and Computer Science is a research institute of the Stichting Mathematisch Centrum, which was founded on February 11, 1946, as a nonprofit institution aiming at the promotion of mathematics, computer science, and their applications. It is sponsored by the Dutch Government through the Netherlands Organization for the Advancement of Pure Research (Z.W.O.).

A PROGRAM FOR SOLVING FIRST KIND FREDHOLM INTEGRAL EQUATIONS BY MEANS OF REGULARIZATION

H.J.J. TE RIELE
Centre for Mathematics and Computer Science, Amsterdam

A program is described for solving a Fredholm integral equation of the first kind with help of the regularization method of Phillips and Tihonov. This type of problems frequently arises in the mathematical analysis of physical problems, like elastic electron-atom scattering.

1980 MATHEMATICS SUBJECT CLASSIFICATION: 65R20, 65V05, 81GXX.
KEY WORDS & PHRASES: Fredholm integral equation of the first kind, regularization, elastic electron-atom scattering, indirect measuring, dispersion relation.
NOTE: This report will be submitted for publication elsewhere.

Report NM-R8416
Centre for Mathematics and Computer Science
P.O. Box 4079, 1009 AB Amsterdam, The Netherlands

0. Program summary

Title of program: F1REGU

Computer: CDC CYBER 175-750; *Installation:* SARA (Academic Computer Centre Amsterdam)

Operating system: NOS/BE

Programming language: FORTRAN 77

Program Library used: NAG FORTRAN LIBRARY, MARK 10. (SUBROUTINE F04ASF; this, in turn, uses other NAG routines)

Programming language: FORTRAN 77

High speed storage requested: 60 K

No. of bits in a word: 60

Overlay structure: none

Other peripherals used: line printer

No. of cards in combined program and test deck: 321

Keywords: Fredholm integral equation of the first kind, regularization, elastic electron-atom scattering, dispersion relation

Nature of the physical problem: Fredholm integral equations of the first kind arise in the mathematical analysis of many physical problems (cf. Nedelkov [3]). An important characteristic of such problems is that the information which we seek about a physical quantity A can only be obtained *indirectly*, by measuring some other quantity B which has some connection with A. Often, this connection can be expressed mathematically in terms of a Fredholm first kind integral equation.

Method of solution The first kind Fredholm integral equation is solved by means of the regularization method of Tihonov ([9, 10, 11]) and Phillips ([4]).

Running time Approximately proportional to the third power of the number of data points.

1. Introduction

The linear first kind Fredholm integral equation

$$\int_a^b K(x,y)f(y)dy = g(x), \; c \leq x \leq d, \tag{1.1}$$

where f is the unknown function, and g and K are given functions, arises in the mathematical analysis of problems from many branches of physics, chemistry and biology ([3]). Also several classical mathematical problems, like the problem of harmonic continuation, numerical inversion of the Laplace transform, the backwards heat equation, and numerical differentiation, can be formulated as equations of the form (1.1).

We assume that f and g are elements of certain linear spaces F and G, respectively. Defining the linear operator $\mathcal{K}:F \to G$ by $(\mathcal{K}f)(x) := \int_a^b K(x,y)f(y)dy$, we write (1.1) in operator notation as:

$$\mathcal{K}f = g, \; g \in G \text{ given}, \; f \in F \text{ sought.} \tag{1.2}$$

In general, numerical solution of (1.1) is difficult, because (1.1) belongs to the class of so-called *ill-posed*, or *improperly posed* problems. The problem (1.2) is ill-posed (in the sense of Hadamard, cf. [2]) if at least one of the following three assertions is *false* (F and G are assumed to be complete metric spaces):

(i) for every $g \in G$ there exists a solution $f \in F$;

(ii) the solution of (1.2) is unique;

(iii) the solution of (1.2) depends continuously on the data g.

Note that this definition depends on the spaces F and G. A problem may be ill-posed with respect to given F and G, but well-posed in other metrics. In general, (1.2) is ill-posed because the solution f of (1.2) does *not* depend continuously on the data function g. This may be explained, at least heuristically, as follows. If K is a smooth function, then $\mathcal{K}$ is a smoothing operator and small perturbations in g may be caused by large perturbations in f, which were smoothed down by $\mathcal{K}$.

In practical situations, the data function g is often the output of some measuring process, so that it is only approximately known in some discrete set of points $x_i \in [c,d]$. Consequently, rather than (1.2) it is more realistic to consider the problem.

$$\mathcal{K}f = \tilde{g} \tag{1.3}$$

where only $\tilde{g}$ and ϵ are known such that $||\tilde{g} - g|| \leq \epsilon$ for some norm $||.||$. This may cause $\tilde{g}$ to lie outside the range of the operator $\mathcal{K}$, so that there may not exist a solution of (1.3).

2. The regularization method.

A survey of numerical methods for solving (1.1) - (1.3) may be found in [5,11]. Here, we describe a simple implementation of the so-called *regularization* method of Phillips ([4]) and Tihonov ([9, 10, 11]). This method essentially consists of the replacement of the ill-posed problem (1.3) by the well-posed problem (i.e. for which the three assertions (i), (ii) and (iii) above are *true*):

Minimize the quadratic functional $\Phi_\alpha(f)$, defined by

$$\Phi_\alpha(f) := ||\mathcal{K}f - \tilde{g}||^2 + \alpha ||Lf||^2, \tag{2.1}$$

over all functions f in the compact set: $\{f : ||\mathcal{K}f - \tilde{g}|| \leq \epsilon\}$.

Here, α is a fixed positive number, the so-called *regularization parameter* and L is some linear operator, e.g., $Lf = f, f'$ or f'', or $Lf = f - \hat{f}$ if an a priori approximation $\hat{f}$ of f can be provided. If Lf is the

i-th derivative of f, then it is customary to speak about i-th *order regularization.*

Under certain, mild conditions, (2.1) has a unique solution, which will be denoted by f_α. Moreover, f_α will converge as $\epsilon \to 0$, uniformly on $[a,b]$, to the solution of the equation $\mathcal{K}f = g$ (if it exists), provided that α satisfies

$$C_1\epsilon^2 < \alpha < C_2\epsilon^2 \tag{2.2}$$

for positive numbers C_1 and C_2. Unfortunately, g is not known exactly and the ill-posedness of (1.3) will, generally, cause the solution f_α of (2.1) to oscillate very wildly around the solution of the equation $\mathcal{K}f = g$, when α is chosen to be close to zero. An increase of α will result in an increase of the residual $||\mathcal{K}f_\alpha - \tilde{g}||$, and a decrease of the "penalty term" $||Lf_\alpha||$; and vice versa. For suitably chosen L the term $||Lf_\alpha||$ will have an increasing damping effect on unwanted oscillations of f_α, with increasing α.

The question then arises: how do we have to choose α? Up till now, this has not been resolved in a satisfactory way. The choice (2.2) may be of some use in practice. In any case, α should be chosen in such a way that *both* $||\mathcal{K}f_\alpha - \tilde{g}||$ *and* $||Lf_\alpha||$ (which, e.g., measures the smoothness of f_α in case $Lf = f''$) are acceptable to the user. Consequently, the proper choice of α depends considerably on the particular problem at hand.

3. The numerical solution of (2.1)

In order to solve (2.1) numerically, we introduce the following descretizations: we assume that $\tilde{g}(x)$ is given in N not necessarily equidistant points $x = x_i$, $i = 1,2,\cdots,N$ $(c \leqslant x_1 < x_2 < \cdots < x_n \leqslant d)$ with $\tilde{g}(x_i) =: g_i$, and we split up the integration interval $[a,b]$ into N subintervals $[y_{j-1}, y_j]$, $j = 1,2,\cdots,N$ $(a = y_0 < y_1 < \cdots < y_N = b)$. The integrals $(\mathcal{K}f)(x)$ occurring in (2.1) are approximated, for any given $x = x_i$, by using the repeated mid-point rule:

$$(\mathcal{K}f)(x_i) = \int_a^b K(x_i,y) f(y) dy = \sum_{j=1}^{N} \int_{y_{j-1}}^{y_j} K(x_i,y) f(y) dy \approx \sum_{j=1}^{N} K_{ij} f_j ,$$

where $K_{ij} := (y_j - y_{j-1}) K(x_i, \bar{y}_j)$, $\bar{y}_j := \frac{1}{2}(y_{j-1} + y_j)$ and $f_j := f(\bar{y}_j)$ is an (unknown) approximation of f in the point $\bar{y}_j$. After defining $\hat{f}_j := \hat{f}(\bar{y}_j)$ as an a priori known estimate of f_j , $\epsilon_i := \sum_{j=1}^{N} K_{ij} f_j - g_i$, $i = 1,2,\cdots,N$, and writing

$$Lf := a_0(f - \hat{f}) + a_1 f' + a_2 f''$$

where $a_i = 0$ or 1, we replace the *continuous* problem (2.1) by the discrete problem:

Minimize the quadratic functional $\bar{\Phi}_\alpha(\vec{f})$, defined by

$$\bar{\Phi}_\alpha(\vec{f}) := \sum_{i=1}^{N} \epsilon_i^2 + \alpha \left\{ a_0 \sum_{j=1}^{N} (f_j - \hat{f}_j)^2 + a_1 \sum_{j=1}^{N-1} (f_{j+1} - f_j)^2 + a_2 \sum_{j=2}^{N-1} (f_{j+1} - 2f_j + f_{j-1})^2 \right\} \tag{3.1}$$

over all vectors $\vec{f} = [f_1, f_2, \cdots, f_N]^T \in \mathbb{R}^N$ for which $\sum_{i=1}^{N} \epsilon_i^2 \leqslant \epsilon^2$.

From the necessary condition $\frac{\partial \bar{\Phi}_\alpha}{\partial f_j} = 0$, $j = 1,2,\cdots,N$, we find, after some simple calculations, the linear matrix-vector equation:

$$\{K^T K + \alpha(a_0 H_0 + a_1 H_1 + a_2 H_2)\}\vec{f} = K^T \vec{g} + \alpha a_0 \vec{\hat{f}} , \tag{3.2}$$

where $\vec{g} = [g_1, \cdots, g_n]^T$, $\vec{\hat{f}} = [\hat{f}_1, \cdots, \hat{f}_N]^T$, $K = (K_{ij})$, $K^T = (K_{ji})$, $H_0 = I_N$ (the $N \times N$ identity matrix),

$$H_1 = \begin{bmatrix} 1 & -1 & & & & \\ -1 & 2 & -1 & & & \\ & -1 & 2 & -1 & & \\ & & \ddots & \ddots & \ddots & \\ & & & -1 & 2 & -1 \\ & & & & -1 & 1 \end{bmatrix}_{N\times N} \quad \text{and } H_2 = \begin{bmatrix} 1 & -2 & 1 & & & & & \\ -2 & 5 & -4 & 1 & & & & \\ 1 & -4 & 6 & -4 & 1 & & & \\ & 1 & -4 & 6 & -4 & 1 & & \\ & & \ddots & \ddots & \ddots & \ddots & \ddots & \\ & & & 1 & -4 & 6 & -4 & 1 \\ & & & & 1 & -4 & 5 & -2 \\ & & & & & 1 & -2 & 1 \end{bmatrix}_{N\times N}$$

The linear symmetric system (3.2) is solved by using the standard NAG-Library routine F04ASF.

4. Description of the program

The main program calls a subroutine F1REGU which solves the minimization problem (3.1). The heading of this subroutine reads as follows:

```
SUBROUTINE F1REGU (KERNEL, N, X, G, Y, ALFA, LINFUN, F, RES)
EXTERNAL KERNEL
REAL KERNEL, X(N), G(N), Y(0:N), RES(6)
```

The parameters of F1REGU are:

KERNEL: a user-supplied external function which delivers the value of the kernel function K in the point (x,y) for any x in the interval $[c,d]$ and any y in the interval $[a,b]$;

N: the number of data points for which g is given and for which approximations to f are to be found; the maximum number allowed is 64;

X(N), G(N): arrays containing, on entry, the abscissae $x_1, \cdots, x_N$ and the corresponding data values $g_1, \cdots, g_N$;

Y(0: N): array of length $N+1$ containing, on entry, a subdivision of $[a,b]$;

ALFA: the regularization parameter, to be supplied by the user;

LINFUN: with this parameter, the user monitors the choice of the linear functional L:
LINFUN $=1$: $Lf = f - \hat{f}$,
$=2$: $Lf = f'$,
$=3$: $Lf = f''$;

F(N): array of length N which, on exit, contains approximations to the solution f in the midpoints $\bar{y}_j$; if LINFUN $=$ 1 then, on entry, the user must provide in F an a priori estimate of the solution in these midpoints;

RES(6): array containing, on exit, the following information: ($||.||$ is the discrete L_2-norm)
RES (1) $= ||\vec{f} - \vec{\hat{f}}||$,
RES (2) $= ||\vec{f}'||$,
RES (3) $= ||\vec{f}''||$,
RES (4) $= ||K\vec{f} - \vec{g}||$,

RES (5) = minimum absolute value of the components of $K\vec{f} - \vec{g}$,
RES (6) = maximum absolute value of the components of $K\vec{f} - \vec{g}$.

5. Workspace

F1REGU uses 8448 words blank common workspace to be declared in the main program as follows:

```
COMMON K(64, 64), MAT (64, 64), RHS (64), WK1 (64), WK2 (64), FH(64)
REAL K, MAT, RHS, WK1, WK2, FH
```

6. Test-examples

The subroutine F1REGU has been tested on the following problem with known solution:

$$K(x,y) = (x+y)^{-1}, \; g(x) = x^{-1} ln \left[\frac{1+x/a}{1+x/b} \right], \; f(y) = y^{-1},$$
$$[a,b] = [c,d] = [1,5], \; N = 16,32,$$
$$x_i = 1+(i-1)*h_1, \; i = 1,2,\cdots,N, \; h_1 = \frac{4}{N-1}, \tag{6.1}$$
$$y_i = 1+i*h_2, \; i = 0,1,\cdots,N, \; h_2 = \frac{4}{N},$$
$$\alpha = 10^{-r}, \; r = 0,1,\cdots,14,$$

For the linear functional L we chose $Lf = f$ (zero-order regularization). The initial vector $\vec{f}$ was taken to be $\vec{0}$. In Table 1 we give for each test combination of α and N the minimum number of correct digits obtained for f in the mid-points $\bar{y}_i = \frac{1}{2}(y_{i-1}+y_i)$, $i=1,2,\cdots,N$. Since the exact solution is monotonic decreasing, we have marked those cases by an asterisk (*) for which the numerical solution was *not* monotonically decreasing.

As a second test we have run the same problem with *perturbed* data g_i, obtained by multiplying $g(x_i)$, $i = 1,2,\cdots,N$, by the factor $1+0.03(2\rho-1)$ where ρ is a random number in the interval (0, 1] generated by the FORTRAN random number generator. Consequently, the maximum perturbation in g is 3%. In order to facilitate reproduction of our tests, we give in Table 2 the perturbed values $\hat{g}(x_i)$ of $g(x_i)$ used in our computations, together with the percentages of the perturbation. The results of the second test are given in the part of Table 1 with heading "Perturbed data". For $\alpha < 10^{-5}$ the numerical values of f obtained were wildly oscillating and completely worthless.

Table 1.
Minimum number of correct digits obtained when
solving problem (6.1) - (1.1) with subroutine F1REGU

	Exact data		Perturbed data	
α	N=16	N=32	N=16	N=32
1	0.1	0.1	0.1	0.1
10^{-1}	0.4	0.4	0.4	0.4
10^{-2}	0.5	0.5	0.4	0.4
10^{-3}	0.9	0.8	0.7	0.7
10^{-4}	1.1	1.1	0.7	0.7
10^{-5}	1.1	1.1	0.1*	-0.2*
10^{-6}	1.6	1.7		
10^{-7}	1.2	1.5		
10^{-8}	1.3	1.7		
10^{-9}	1.1	1.7		
10^{-10}	1.0	1.6		
10^{-11}	1.0	1.5		
10^{-12}	0.7*	1.3		
10^{-13}	0.7*	1.0*		
10^{-14}	0.4*	0.1*		

* : numerical solution *not* monotonically decreasing

In the case of exact data, the best results were obtained for values of α which lie in the range $10^{-11} < \alpha < 10^{-4}$. Doubling the number N of discretization points has some effect only for very small values of $\alpha(< 10^{-8}$, say).

In the case of inexact data, the best results were obtained for $\alpha = 10^{-3}$ and $\alpha = 10^{-4}$. A maximal error of 3% corresponds, roughly, to $\epsilon = 0.03$ in (2.1). The values of α for which the best results were obtained agree reasonably well with the theoretical choice $\alpha = \mathcal{O}(\epsilon^2)$ expressed in (2.2).

In Figures 1 and 2 we present graphs of the numerical solutions f_α obtained in the cases $\alpha = 10^{-3}, 10^{-4}, N=16, 32$, with inexact data. The drawn line is the exact solution.

The source-text of F1REGU together with the line printer output of the tests shown in Figures 1 and 2 is given in the APPENDIX.

The subroutine F1REGU has also been used recently to solve a problem arising in elastic electron-atom scattering ([8,12]). Some experiments with a (ALGOL 60) predecessor of F1REGU have been reported in [5].

Table 2
Perturbed values $\hat{g}(x_i)$ of $g(x_i)$, used in the test examples.

N=16

x_i	$g(x_i)$	$\hat{g}(x_i)$	error (%)
1.0000	.510826	.500345	-2.1
1.2667	.467766	.476891	2.0
1.5333	.431776	.427548	-1.0
1.8000	.401186	.409976	2.2
2.0667	.374826	.365242	-2.6
2.3333	.351849	.352851	.3
2.6000	.331624	.333124	.5
2.8667	.313673	.316543	.9
3.1333	.297623	.301847	1.4
3.4000	.283180	.283277	.0
3.6667	.270109	.278030	2.9
3.9333	.258219	.263798	2.2
4.2000	.247355	.241919	-2.2
4.4667	.237387	.239189	.8
4.7333	.228207	.232738	2.0
5.0000	.219722	.217096	-1.2

N=32

x_i	$g(x_i)$	$\hat{g}(x_i)$	error (%)
1.0000	.510826	.500345	-2.1
1.1290	.488975	.498514	2.0
1.2581	.469034	.464441	-1.0
1.3871	.450752	.460627	2.2
1.5161	.433920	.422825	-2.6
1.6452	.418367	.419558	.3
1.7742	.403946	.405773	.5
1.9032	.390533	.394106	.9
2.0323	.378022	.383387	1.4
2.1613	.366322	.366448	.0
2.2903	.355354	.365775	2.9
2.4194	.345050	.352504	2.2
2.5484	.335348	.327978	-2.2
2.6774	.326197	.328672	.8
2.8065	.317549	.323854	2.0
2.9355	.309362	.305665	-1.2
3.0645	.301600	.305834	1.4
3.1935	.294230	.293497	-.2
3.3226	.287222	.284772	-.9
3.4516	.280549	.283210	.9
3.5806	.274188	.273713	-.2
3.7097	.268116	.272964	1.8
3.8387	.262313	.269326	2.7
3.9677	.256763	.258344	.6
4.0968	.251448	.252259	.3
4.2258	.246354	.240569	-2.3
4.3548	.241466	.237938	-1.5
4.4839	.236772	.236306	-.2
4.6129	.232261	.228952	-1.4
4.7419	.227923	.228027	.0
4.8710	.223746	.225888	1.0
5.0000	.219722	.217514	-1.0

References

[1] J.J.M. Cuppen, *Regularization methods and parameter estimation methods for the solution of Fredholm integral equations of the first kind,* pp. 67-85 in [6].

[2] M.M. Lavrentiev, *Some improperly posed problems of mathematical physics,* Springer Tracts in natural philosophy, vol. 11, Springer, Berlin, 1967.

[3] I.P. Nedelkov, *Improper problems in computational physics,* Computer Physics Comm. 4 (1972) 157-164.

[4] D.L. PHILLIPS, *A technique for the numerical solution of certain integral equations of the first kind*, J. ACM 9 (1962) 84-97.

[5] H.J.J. te RIELE, *Regularisatiemethoden voor integraalvergelijkingen van de eerste soort*, pp. 147-176 in: H.J.J. te Riele (ed.), Colloquium Numerieke Programmatuur, Deel 2, MC Syllabus 29.2, Amsterdam 1977 (in Dutch).

[6] H.J.J. te RIELE, (ed.) *Colloquium numerical treatment of integral equations*, MC Syllabus 41, Math. Centre, Amsterdam, 1979.

[7] H.J.J. te RIELE, *Numerical methods for first kind Fredholm integral equations*, pp. 47-65 in [6].

[8] H.J.J. te RIELE, and R. WAGENAAR, *Numerical solution of a first kind Fredholm integral equation arising in electron-atom scattering*, in: G. Hämmerlin and K.-H. Hoffmann (eds.), Constructive methods for the practical treatment of integral equations, Proceedings of a Conference at Oberwolfach (B.R.D) June 24-30, 1984, to appear.

[9] A.N. TIHONOV, *Solution of incorrectly formulated problems and the regularization method*, Soviet Math. Dokl. 4 (1963) 1035-1038.

[10] A.N. TIHONOV, *Regularization of incorrectly posed problems*, Soviet Math. Dokl. 4 (1963) 1624-1627.

[11] A.N. TIHONOV and V.Y. ARSENIN, *Solution of ill-posed problems*, V.H. Winston and Sons, Washington D.C., 1977.

[12] R. WAGENAAR, *Small angle elastic scattering of electrons by noble gas atoms*, Doctor's Thesis, Amsterdam, 1984.



Figure 1



Figure 2

Appendix

A.1 The source-text of SUBROUTINE F1REGU

```
      SUBROUTINE F1REGU(KERNEL, N, X, G, Y, ALFA, LINFUN, F, RES)
      EXTERNAL KERNEL
      COMMON K(64,64), MAT(64,64), RHS(64), WK1(64), WK2(64), FH(64)
      REAL KERNEL, X(N), G(N), Y(Ø:N), F(N), RES(6),
     +K, MAT, RHS, WK1, WK2, FH
C-----------------------------------------------------------------------
C PROBLEM DESCRIPTION
C-----------------------------------------------------------------------
C
C F1REGU  SOLVES A LINEAR FREDHOLM INTEGRAL EQUATION  OF  THE FIRST KIND
C BY USING THE REGULARIZATION METHOD OF PHILLIPS AND TIHONOV.
C REFERENCE: H.J.J. TE RIELE "REGULARISATIEMETHODEN VOOR INTEGRAALVERGE-
C LIJKINGEN VAN DE EERSTE SOORYT", PP. 147 - 176 IN:
C COLLOQUIUM NUMERIEKE PROGRAMMATUUR, II, MC SYLL. 29.2, AMSTERDAM 1977.
C
C THE FREDHOLM FIRST KIND INTEGRAL EQUATION READS AS FOLLOWS:
C
C          _Y(N)
C         /\
C        /
C       /  KERNEL(X,Y)F(Y)DY = G(X),  X(1) <= X <= X(N),
C    \_/
C   Y(Ø)
C
C WHERE KERNEL AND G ARE GIVEN FUNCTIONS, AND  F  IS  THE  UNKNOWN
C FUNCTION TO BE FOUND. THE FUNCTION G IS KNOWN ONLY  APPROXIMATELY
C IN A DISCRETE SET OF POINTS.
C-----------------------------------------------------------------------
C THE PARAMETERS OF F1REGU
C-----------------------------------------------------
C KERNEL A USER-SUPPLIED EXTERNAL FUNCTION WHICH DELIVERS THE
C        VALUE OF THE KERNEL FUNCTION IN THE POINT (X,Y),
C        FOR ANY X  IN THE INTERVAL X(1)  <= X  <= X(N)
C        AND ANY Y  IN THE INTERVAL Y(Ø)  <= Y  <= Y(N).
C N      THE NUMBER OF POINTS FOR WHICH G IS  GIVEN  AND
C        FOR WHICH F1REGU FINDS APPROXIMATIONS TO F;
C        THE MAXIMUM ALLOWED VALUE OF N IS 64.
C X(N) &   ARRAYS CONTAINING THE ABSCISSAE X(1),...,X(N)  AND
C G(N)     THE CORRESPONDING VALUES  G(1), ..., G(N),
C          TO BE SUPPLIED BY THE USER.
C Y(Ø:N) ARRAY OF LENGTH N+1 CONTAINING A SUBDIVISION OF THE
C        INTERVAL OF INTEGRATION  (Y(Ø),Y(N))  ON  WHICH THE
C        UNKNOWN FUNCTION F HAS TO BE FOUND. TO BE SUPPLIED
C        BY THE USER.
C ALFA   THE REGULARIZATION PARAMETER, THE VALUE OF WHICH HAS
C        TO BE SUPPLIED BY THE USER.
```

```
LINFUN WITH THIS PARAMETER, THE USER  PROVIDES  INFORMATION
       ABOUT THE CHOICE OF THE LINEAR FUNCTIONAL L  IN  THE
       REGULARIZATION METHOD.
       LINFUN = 1: L = F - FH WHERE FH IS AN A PRIORI
                       ESTIMATE OF THE SOLUTION F, TO  BE
                       PROVIDED BY THE USER
       LINFUN = 2: L = F' (THE DERIVATIVE OF F)
       LINFUN = 3: L = F'' (THE SECOND DERIVATIVE OF F)
F(N)   ARRAY OF LENGTH N WHICH, ON EXIT, CONTAINS  APPROXI-
       MATIONS  OF THE  SOLUTION   F   IN  THE   MID-POINTS
       (Y(I-1)+Y(I))/2, FOR I = 1,2, ... , N.
       IF LINFUN = 1 THEN, ON ENTRY, THE USER MUST  PROVIDE
       AN A PRIORI ESTIMATE OF THE SOLUTION IN  THESE  MID-
       POINTS.
RES(6) ARRAY CONTAINING,ON EXIT, THE FOLLOWING INFORMATION:
       RES(1)=NORM(F-FH) (FOR FH, SEE LINFUN)
       RES(2)=NORM(F')
       RES(3)=NORM(F'')
       RES(4)=NORM(K*F-G)
       RES(5)=MIN. ABS. COMPONENT VALUE OF K*F-G
       RES(6)=MAX. ABS. COMPONENT VALUE OF K*F-G
       HERE, NORM IS THE DISCRETE L2 - NORM.
----------------------------------------------------------

WORKING SPACE
----------------------------------------------------------
BLANK COMMON BLOCKS
K(64,64), MAT(64,64),  RHS(64), WK1(64), WK2(64),   FH(64)
WHICH  REFLECT THE MAXIMUM VALUE OF N ALLOWED (64).
----------------------------------------------------------
-----------------
SOME PREPARATIONS
-----------------
    NM1=N-1
    NM2=N-2
    ALFA2=2.*ALFA
    ALFA4=4.*ALFA
    ALFA5=5.*ALFA
    ALFA6=6.*ALFA
    IF(LINFUN.EQ.1)THEN
    DO 1 I=1,N
  1 FH(I)=F(I)
    ELSE
    DO 2 I=1,N
  2 FH(I)=Ø.
    END IF
---------------------
FILL THE ARRAY K(N,N)
---------------------
    DO 1Ø I=1, N
    DO 1Ø J=1, N
 1Ø K(I,J)=(Y(J)-Y(J-1))*KERNEL(X(I),(Y(J)+Y(J-1))/2.)
```

```
---------------------------------------
FILL THE UPPER TRIANGLE OF MAT WITH K'K
---------------------------------------
    DO 30 I=1, N
    DO 30 J=I, N
    H=0
    DO 20 L=1, N
 20 H=H+K(L,I)*K(L,J)
 30 MAT(I,J)=H
---------------------------------------------
IF LINFUN=1 ADD ALFA*(THE UNIT MATRIX) TO MAT
---------------------------------------------
    IF(LINFUN.EQ.1)THEN
    DO 40 I=1,N
 40 MAT(I,I)=MAT(I,I)+ALFA
    END IF
--------------------------------------------------------------------
IF LINFUN=2 ADD ALFA*H1 TO THE UPPER TRIANGLE OF MAT, WHERE H1 IS A
SPECIAL FIRST DIFFERENCE MATRIX
--------------------------------------------------------------------
    IF(LINFUN.EQ.2)THEN
    MAT(1,1)=MAT(1,1)+ALFA
    MAT(1,2)=MAT(1,2)-ALFA
    MAT(N,N)=MAT(N,N)+ALFA
    DO 50 I=2,NM1
    MAT(I,I)=MAT(I,I)+ALFA2
 50 MAT(I,I+1)=MAT(I,I+1)-ALFA
    END IF
--------------------------------------------------------------------
IF LINFUN=3 ADD ALFA*H2 TO THE UPPER TRIANGLE OF MAT, WHERE H2 IS A
SPECIAL SECOND DIFFERENCE MATRIX
--------------------------------------------------------------------
    IF(LINFUN.EQ.3)THEN
    MAT(1,1)=MAT(1,1)+ALFA
    MAT(1,2)=MAT(1,2)-ALFA2
    MAT(1,3)=MAT(1,3)+ALFA
    MAT(2,2)=MAT(2,2)+ALFA5
    MAT(2,3)=MAT(2,3)-ALFA4
    MAT(2,4)=MAT(2,4)+ALFA
    MAT(NM1,NM1)=MAT(NM1,NM1)+ALFA5
    MAT(NM1,N)=MAT(NM1,N)-ALFA2
    MAT(N,N)=MAT(N,N)+ALFA
    DO 60 I=3,NM2
    MAT(I,I)=MAT(I,I)+ALFA6
    MAT(I,I+1)=MAT(I,I+1)-ALFA4
 60 MAT(I,I+2)=MAT(I,I+2)+ALFA
    END IF
----------------------------------
FILL THE ARRAY RHS(N) WITH K'G
----------------------------------
    DO 90 I=1, N
    H=0.
    DO 80 L=1, N
 80 H=H+K(L,I)*G(L)
 90 RHS(I)=H
```

```
-----------------------------------------------------------------------
IF LINFUN=1 ADD ALFA*F TO RHS, WHERE F CONTAINS A FIRST ESTIMATE
OF THE SOLUTION F
-----------------------------------------------------------------------
    IF(LINFUN.EQ.1)THEN
    DO 100 I=1,N
100 RHS(I)=RHS(I)+ALFA*F(I)
    END IF
-------------------------------------------
NOW SOLVE THE LINEAR SYSTEM MAT * F = RHS
-------------------------------------------
    IFAIL=0
    CALL F04ASF(MAT,64,RHS,N,F,WK1,WK2,IFAIL)
----------------
COMPUTE RESIDUES
------------------------------
FIRST RES(1), RES(2) AND RES(3)
------------------------------
    H1=(F(1)-FH(1))**2 + (F(N)-FH(N))**2
    H2=(F(2)-F(1))**2
    H3=0.
    DO 110 I=2,NM1
    H1=H1+(F(I)-FH(I))**2
    H2=H2+(F(I+1)-F(I))**2
    H3=H3+(F(I+1)-2.*F(I)+F(I-1))**2
110 CONTINUE
    RES(1)=SQRT(H1)
    RES(2)=SQRT(H2)
    RES(3)=SQRT(H3)
------------------------------
NEXT RES(4), RES(5) AND RES(6)
------------------------------
    RESID=0.
    RESMAX=0.
    RESMIN=1.E100
    DO 130 I=1,N
    H=0.
    DO 120 J=1,N
120 H=H+K(I,J)*F(J)
    H=H-G(I)
    RESID=RESID+H*H
    EPS=ABS(H)
    IF(EPS.GT.RESMAX)RESMAX=EPS
    IF(EPS.LT.RESMIN)RESMIN=EPS
130 CONTINUE
    RES(4)=SQRT(RESID)
    RES(5)=RESMIN
    RES(6)=RESMAX
    RETURN
    END
```

A.2 Test run output

TEST OF F1REGU
ORDER OF REGULARIZATION= 0
NUMBER OF POINTS N=16 ALFA= .00010000
RESIDUES RES(1),...,RES(6)=
.1763E+01 .1416E+00 .1100E-01 .2256E-01 .1611E-04 .1103E-01

Y	F(Y) EXACT	F(Y) COMPUTED	NUMBER OF CORRECT DIGITS	ERROR PERCENTAGE
1.125000	.888889	.719862	.7	19.0
1.375000	.727273	.666015	1.1	8.4
1.625000	.615385	.611811	2.2	.6
1.875000	.533333	.560268	1.3	-5.1
2.125000	.470588	.512573	1.0	-8.9
2.375000	.421053	.469044	.9	-11.4
2.625000	.380952	.429594	.9	-12.8
2.875000	.347826	.393959	.9	-13.3
3.125000	.320000	.361803	.9	-13.1
3.375000	.296296	.332782	.9	-12.3
3.625000	.275862	.306564	1.0	-11.1
3.875000	.258065	.282844	1.0	-9.6
4.125000	.242424	.261347	1.1	-7.8
4.375000	.228571	.241827	1.2	-5.8
4.625000	.216216	.224069	1.4	-3.6
4.875000	.205128	.207880	1.9	-1.3

MINIMUM NUMBER OF CORRECT DIGITS: .7

TEST OF F1REGU
ORDER OF REGULARIZATION= 0
NUMBER OF POINTS N=16 ALFA= .00100000
RESIDUES RES(1),...,RES(6)=
.1751E+01 .1367E+00 .1952E-01 .2297E-01 .3586E-03 .1283E-01

Y	F(Y) EXACT	F(Y) COMPUTED	NUMBER OF CORRECT DIGITS	ERROR PERCENTAGE
1.125000	.888889	.712591	.7	19.8
1.375000	.727273	.641722	.9	11.8
1.625000	.615385	.582561	1.3	5.3
1.875000	.533333	.532537	2.8	.1
2.125000	.470588	.489759	1.4	-4.1
2.375000	.421053	.452814	1.1	-7.5
2.625000	.380952	.420627	1.0	-10.4
2.875000	.347826	.392367	.9	-12.8
3.125000	.320000	.367380	.8	-14.8
3.375000	.296296	.345148	.8	-16.5
3.625000	.275862	.325257	.7	-17.9
3.875000	.258065	.307366	.7	-19.1
4.125000	.242424	.291201	.7	-20.1
4.375000	.228571	.276530	.7	-21.0
4.625000	.216216	.263164	.7	-21.7
4.875000	.205128	.250941	.7	-22.3

MINIMUM NUMBER OF CORRECT DIGITS: .7

TEST OF F1REGU
ORDER OF REGULARIZATION= 0
NUMBER OF POINTS N=32 ALFA= .00010000
RESIDUES RES(1),...,RES(6)=
.2520E+01 .1196E+00 .4719E-02 .3017E-01 .3131E-03 .1149E-01

Y	F(Y) EXACT	F(Y) COMPUTED	NUMBER OF CORRECT DIGITS	ERROR PERCENTAGE
1.062500	.941176	.781930	.8	16.9
1.187500	.842105	.751169	1.0	10.8
1.312500	.761905	.719359	1.3	5.6
1.437500	.695652	.687305	1.9	1.2
1.562500	.640000	.655555	1.6	-2.4
1.687500	.592593	.624474	1.3	-5.4
1.812500	.551724	.594304	1.1	-7.7
1.937500	.516129	.565195	1.0	-9.5
2.062500	.484848	.537235	1.0	-10.8
2.187500	.457143	.510467	.9	-11.7
2.312500	.432432	.484903	.9	-12.1
2.437500	.410256	.460534	.9	-12.3
2.562500	.390244	.437333	.9	-12.1
2.687500	.372093	.415267	.9	-11.6
2.812500	.355556	.394294	1.0	-10.9
2.937500	.340426	.374369	1.0	-10.0
3.062500	.326531	.355443	1.1	-8.9
3.187500	.313725	.337470	1.1	-7.6
3.312500	.301887	.320402	1.2	-6.1
3.437500	.290909	.304191	1.3	-4.6
3.562500	.280702	.288792	1.5	-2.9
3.687500	.271186	.274162	2.0	-1.1
3.812500	.262295	.260259	2.1	.8
3.937500	.253968	.247042	1.6	2.7
4.062500	.246154	.234474	1.3	4.7
4.187500	.238806	.222519	1.2	6.8
4.312500	.231884	.211143	1.0	8.9
4.437500	.225352	.200314	1.0	11.1
4.562500	.219178	.190001	.9	13.3
4.687500	.213333	.180177	.8	15.5
4.812500	.207792	.170815	.7	17.8
4.937500	.202532	.161889	.7	20.1

MINIMUM NUMBER OF CORRECT DIGITS: .7

TEST OF F1REGU
ORDER OF REGULARIZATION= 0
NUMBER OF POINTS N=32   ALFA= .00100000
RESIDUES RES(1),...,RES(6)=
.2486E+01 .1138E+00 .9113E-02 .3137E-01 .2840E-04 .1277E-01

Y	F(Y) EXACT	F(Y) COMPUTED	NUMBER OF CORRECT DIGITS	ERROR PERCENTAGE
1.062500	.941176	.768431	.7	18.4
1.187500	.842105	.723721	.9	14.1
1.312500	.761905	.683226	1.0	10.3
1.437500	.695652	.646404	1.2	7.1
1.562500	.640000	.612803	1.4	4.2
1.687500	.592593	.582039	1.7	1.8
1.812500	.551724	.553787	2.4	-.4
1.937500	.516129	.527766	1.6	-2.3
2.062500	.484848	.503736	1.4	-3.9
2.187500	.457143	.481492	1.3	-5.3
2.312500	.432432	.460850	1.2	-6.6
2.437500	.410256	.441655	1.1	-7.7
2.562500	.390244	.423769	1.1	-8.6
2.687500	.372093	.407068	1.0	-9.4
2.812500	.355556	.391447	1.0	-10.1
2.937500	.340426	.376811	1.0	-10.7
3.062500	.326531	.363074	1.0	-11.2
3.187500	.313725	.350161	.9	-11.6
3.312500	.301887	.338005	.9	-12.0
3.437500	.290909	.326545	.9	-12.2
3.562500	.280702	.315727	.9	-12.5
3.687500	.271186	.305501	.9	-12.7
3.812500	.262295	.295824	.9	-12.8
3.937500	.253968	.286655	.9	-12.9
4.062500	.246154	.277957	.9	-12.9
4.187500	.238806	.269698	.9	-12.9
4.312500	.231884	.261847	.9	-12.9
4.437500	.225352	.254377	.9	-12.9
4.562500	.219178	.247262	.9	-12.8
4.687500	.213333	.240480	.9	-12.7
4.812500	.207792	.234008	.9	-12.6
4.937500	.202532	.227828	.9	-12.5

MINIMUM NUMBER OF CORRECT DIGITS: .7